

SUMIF 関数

SUMIF 関数

文字通り SUM(合計) + IF(条件) です。

条件に適合しているもののみを集計対象にして合計を求めるための関数です。

検索範囲で検索条件と一致する行の合計範囲の値を合計します。

=SUMIF(検索範囲,検索条件,合計範囲)

SUMIFS 関数

SUMIFS 関数は Excel2007 で追加された比較的新しい関数で、複合条件のすべてを満足するものの合計値を求めるものです。

SUMIF と違って、最初に合計範囲を記述し、次から、検索範囲 1、検索条件 1、検索範囲 2、検索条件 2・・・という風に続けます。

=SUMIFS(合計範囲,検索範囲 1,検索条件 1,[検索範囲 2,検索条件 2]・・・)

COUNTIF 関数

COUNT(個数) + IF(条件) を意味している COUNTIF は、条件に適合しているもののセルの個数を求めるための関数です。

検索範囲で検索条件と一致する行の合計範囲の値を合計します。

=COUNTIF(検索範囲,検索条件,合計範囲)

SUMIF 関数の具体例

下の 2 表は支店別・商品別の売上表(上の表)を元に、全支店トータルの商品別売上表(下の表)を算出しているものです。

	A	B	C	D	E
1		商品名	4月売上	5月売上	6月売上
2	A支店	商品A	1,500	1,250	2,000
3		商品B	2,350	2,200	2,950
4		商品C	10,000	9,000	11,500
5	B支店	商品A	1,500	1,750	1,300
6		商品C	7,000	7,500	6,000
7	C支店	商品A	2,300	2,000	1,800
8		商品B	6,900	7,300	7,700
9		合計	31,550	31,000	33,250
10					
11		商品名	4月売上	5月売上	6月売上
12	全支店	商品A	5,300	5,000	5,100
13		商品B	9,250	9,500	10,650
14		商品C	17,000	16,500	17,500
15		合計	31,550	31,000	33,250

この場合、例えば

C12 のセルには「=SUM(C2,C5,C7)」 (or「=C2+C5+C7」)、

C13 のセルは「=SUM(C3,C8)」、

C14 のセルは「=SUM(C4,C6)」

という式を用いている方が多いのではないのでしょうか・・・。

上述のような比較的小さい表であればまだしもですが、この方法では大きな表になるほど加算対象にすべきセルが漏れる恐れが増します。また新たに支店が増えた場合など、それぞれの式をすべて見直して直す必要が出てきます。

SUMIF を用いると以下のようにすることができます。

C 列

D 列

行 12 =SUMIF(\$B\$2:\$B\$8,\$B12,C\$2:C\$8) =SUMIF(\$B\$2:\$B\$8,\$B12,D\$2:D\$8)

行 13 =SUMIF(\$B\$2:\$B\$8,\$B13,C\$2:C\$8) =SUMIF(\$B\$2:\$B\$8,\$B13,D\$2:D\$8)

セル C12 の関数は、セル B2 から B8 までを検索範囲として、検索条件としてセル B12 と合致するもののみを加算対象とし、セル C2 から C8 までの範囲で加算計算する、という意味ですね。

セル B12 は「商品 A」ですから、セル B2 から B8 までの中で「商品 A」という文字列を含む行のみを対象としますので、実際の加算は C2、C5、C7 のセルということになります。

このように加算対象の項目を条件式で規定すれば、対象漏れということはありません。

また上記を良く見て頂くと、関数の中に絶対参照と複合参照が混ざっていて、C12のセルをドラッグで右方向・下方向にコピーするだけで、残りのセルに適切な数式が埋まることがお分かりだと思います。

絶対参照・複合参照を使いこなすことが表計算の醍醐味といえましょう。

【便利知識】

SUMIF は次のように検索条件として検査値そのものを指定することができます。

中でも「スペースでない行を集計対象とする」という方法を覚えておくと便利だと思います。

(例 1) =SUMIF(A2:A20,"小計",B2:B20)

A2~A20 の中で"小計"となっている行の B 列の値を集計し、合計を求める

(例 2) =SUMIF(A2:A15,"*",B2:B15)

A2~A15 の中で"スペースでない"行の B 列の値の合計を求める

(例 3) =SUMIF(A2:A15,"*償却*",B2:B15)

A2~A15 の中で"償却"という文字列を含む行の B 列の値の合計を求める

【便利知識】

文字列の箇所のアスタリスク "*" は、ワイルドカードの一種で、0 文字以上の何らかの文字列を表します。

"千葉県*" 千葉県で始まる文字列

"*山" 山で終わる文字列

"*償却*" 償却を含む文字列(償却で始まる、償却で終わる、文字列を含む)

クエスチオン "?" は、1 文字分のワイルドカードです。

"川?" 川で始まる 2 文字の文字列

"???川" 川で終わる 4 文字の文字列